

INSTRUCTIONS FOR LENGTH-WEIGHT PROGRAMS for
IBM 1620 in FORTRAN—FORMAT (FORTRAN I)

Wayne E. Swingle

Zoology-Entomology Department Series
Fisheries No. 1

AGRICULTURAL EXPERIMENT STATION
AUBURN UNIVERSITY

E. V. Smith, Director,

September 1964

Auburn, Alabama

Instructions for Length-Weight Programs

for

IBM 1620

in

Fortran/Format (Fortran I)

by

Wayne E. Swingle*

INTRODUCTION

The need for tables and equations expressing the relationship between length and weight of a species of fish arises frequently in fisheries work. At certain times, it is impractical to take measurements of both length and weight since the apparatus used for weighing is not accurate enough to weigh the smaller specimens. Also, errors in recording or weighing under field conditions occasionally make it necessary to substitute estimated weights for field data. In such cases, tables expressing the average weight for a given length are invaluable.

Equations relating length and weight are useful for comparing populations, for examining growth differences of fish subjected to different treatment effects, for calculating condition factors, and for determining Von Bertalanffy growth equations in terms of weight.

The author, through this set of computer programs, has attempted to remove some of the tedious labor associated with the tabulation of length-weight relationships. He has attempted to make this set of programs as versatile as possible. Because the present trend is toward reporting length-weight relationships in inches and pounds, the programs are designed to convert other units of measure to inches and pounds. However, the programs will compute equations and empirical

*From a thesis submitted as partial requirement for M. S. degree in Fisheries Management.

relations in terms of other units provided the specifications given in the subsequent sections are followed.

Program I computes two length-weight equations, Program II tabulates empirical relationships including average weight, maximum weight, minimum weight and the condition index. Program III transforms data cards in units other than inches and pounds to data in inches and pounds. Detailed descriptions of the programs are given in the sections that follow.

DESCRIPTIONS OF PROGRAMS

Program I

Title: Length-Weight Relationships of Fishes I, IBM 1620, Fortran/Format

This program computes the standard length-weight equations and also a third degree polynomial equation by least-squares procedures as outlined by Hader and Grandage (1958). Input may include measurement data of individual specimens or of many specimens combined in class intervals for which the average is computed and considered as one observation. The output consists of: a) estimates of the parameters of the three equations,

$$\log (W) = \log (a) + b \log (L) \quad (1)$$

$$W = aL^b \quad (2)$$

$$W = b_1 + b_2L + b_3L^2 + b_4L^3 \quad (3)$$

b) for equation (1), the standard error, correlation coefficient, means, intercept, sums of squares of the observations and deviations, and the numbers of fish and observations (one observation may include many fish where class intervals are used); c) for equation (3), the X'X matrix of the normal equations, the forward abbreviated Doolittle solution of this matrix, and the inverse of the X'X matrix; d) estimated weights by 1-inch intervals using equations (2) and (3).

The program utilizes 1600 storage positions. Data cards are processed at the rate of 37 per minute with calculations and punch out requiring an additional 3 minutes. Estimates of the parameters are accurate to at least four significant

figures.

Options: Measurement data may be in inches-pounds, inches-ounces, inches-grams, millimeters-pounds and millimeters-grams; output is in inches-pounds, or output may be in same units as measurement data. Computation of the third degree polynomial is under control of a header card and may be omitted. Data of mixed systems of measurement may be entered and a single equation with output in inches-pounds is computed.

Restrictions: Input and output are on cards with one data card required for each observation. Input is limited to length measurements up to 5 digits (or 4 digits and a decimal), weight measurements up to 10 digits, and 8 digits are available for number of fish per class interval (where class intervals are used the weight must be the total weight of all fish in that interval). Estimated weights from the fitted equations are limited to 99 length-integer intervals (for inches, weights may be obtained for fish 1 to 99 inches by one-inch intervals).

Source Program

```

1  SUMX2=0
   SUMY2=0
   SUMY=0
   SUMX=0
   SUMXY=0
   TOTN=0
   SFNO=0
   SWY=0
   SWX=0
   SWX2=0
   SWX3=0
   SWX4=0
   SWX5=0
   SWX6=0
   SWXY=0
   SWX2Y=0
   SWX3Y=0
200 READ5,MSET,LSET,JSET
    5 FORMAT(I2,3X,I1,1X,I1)
    2 READ 99,SIZE,FNO,TOTWT,PBNO,KI
    99 FORMAT(F5.2,F8.0,F10.5,F4.0,44X,I1)
       IF(SIZE-8888.)20,20,14
    20 IF(KI-1)21,21,200
    21 IF(TOTWT)23,2,23
    23 IF(SIZE)25,2,25
    25 IF(FNO)22,2,22
    22 AVGWT=TOTWT/FNO

```

```

IF(LSET-1)105,105,100
100 IF(LSET-3)106,107,101
101 IF(LSET-5)108,109,105
105 WY=AVGWT
WX=SIZE
GO TO 6
106 WY=AVGWT/453.59
WX=SIZE/25.4
GO TO 6
107 WY=AVGWT
WX=SIZE/25.4
GO TO 6
108 WY=AVGWT/453.59
WX=SIZE
GO TO 6
109 WY=AVGWT/16.0
WX=SIZE
GO TO 6
6 Y=LOG(WY)*.43429448
X=LOG(WX)*.43429448
SUMX2=SUMX2+X*X
SUMX=SUMX+X
SUMY2=SUMY2+Y*Y
SUMY=SUMY+Y
SUMXY=SUMXY+X*Y
TOTN=TOTN+1.
SFNO=SFNO+FNO
SWX=SWX+WX
SWY=SWY+WY
SWX2=SWX2+WX*WX
SWX3=SWX3+WX*WX*WX
SWX4=SWX4+WX*WX*WX*WX
SWX5=SWX5+WX*WX*WX*WX*WX
SWX6=SWX6+WX*WX*WX*WX*WX*WX
SWXY=SWXY+WX*WY
SWX2Y=SWX2Y+WX*WX*WY
SWX3Y=SWX3Y+WX*WX*WX*WY
GO TO 2
14 CTX=SUMX**2/TOTN
CTY=SUMY**2/TOTN
CTXY=(SUMX*SUMY)/TOTN
SSX2=SUMX2-CTX
SSY2=SUMY2-CTY
SSXY=SUMXY-CTXY
B=SSXY/SSX2
AVGX=SUMX/TOTN
AVGY=SUMY/TOTN
ALOG=AVGY-B*AVGX
A=10.**ALOG
R=SSXY/SQRT(SSX2*SSY2)
S=SQRT((SSY2-(SSXY*SSXY)/SSX2)/(TOTN-2.))
SD=SSY2-(SSXY*SSXY)/SSX2
PUNCH85,SUMX,SUMX2,SUMXY
PUNCH4
85 FORMAT(3HSX=,E12.6,4HSX2=,E12.6,4HSXY=,E12.6)
PUNCH 86,SUMY,SUMY2
86 FORMAT(3HSY=,E12.6,4HSY2=,E12.6)
PUNCH87,SD
87 FORMAT(3HSD=,E12.6)

```

```
PUNCH 13,R,AVGX,AVGY
13 FORMAT(10H CORR COEF=,F10.6,5HAVGX=,F12.8,5HAVGY=,F12.8)
PUNCH4
PUNCH 96,SFNO,PBNO,SSX2,SSY2
96 FORMAT(5HSFNO=,F10.0,5HPBNO=,F6.0,5HSSX2=,F12.6,5HSSY2=,F12.6)
PUNCH4
PUNCH97,TOTN,SSXY
97 FORMAT(5HTOTN=,F8.0,5HSSXY=,F12.6)
PUNCH4
PUNCH88,B,A,ALOG
88 FORMAT(2HB=,F10.5,2HA=,F12.8,5HALOG=,F12.5)
PUNCH4
PUNCH 44,S
44 FORMAT(10HSTD ERROR=,F12.8)
PUNCH4
PUNCH34,SWX,SWX2,SWX3,SWX4
PUNCH4
34 FORMAT(5HSUMX=,E12.6,6HSUMX2=,E12.6,6HSUMX3=,E12.6,6HSUMX4=,E12.6)
35 FORMAT(6HSUMX5=,E12.6,6HSUMX6=,E12.6,5HSUMY=,E12.6)
PUNCH35,SWX5,SWX6,SWY
PUNCH4
PUNCH36,SWXY,SWX2Y,SWX3Y
PUNCH4
36 FORMAT(6HSUMXY=,E12.6,7HSUMX2Y=,E12.6,7HSUMX3Y=,E12.6)
IF(JSET-1)37,37,38
38 B1=0
B2=0
B3=0
B4=0
GO TO 76
37 A11=TOTN
A12=SWX
A13=SWX2
A14=SWX3
G11=SWY
B11=TOTN/TOTN
B12 = SWX/TOTN
B13=SWX2/TOTN
B14=SWX3/TOTN
G12=SWY/TOTN
A22=SWX2-A12*B12
A23=SWX3-A12*B13
A24=SWX4-A12*B14
G21=SWXY-A12*G12
B22=A22/A22
B23=A23/A22
B24=A24/A22
G22=G21/A22
A33=SWX4-A13*B13-A23*B23
A34=SWX5-A13*B14-A23*B24
G31=SWX2Y-A13*G12-A23*G22
B33=A33/A33
B34=A34/A33
G32=G31/A33
A44=SWX6-A14*B14-A24*B24-A34*B34
G41=SWX3Y-A14*G12-A24*G22-A34*G32
B44=A44/A44
G42=G41/A44
PUNCH39
```

```
39 FORMAT(26HDOOLITTLE FORWARD SOLUTION)
PUNCH40,A11,A12,A13,A14,G11
40 FORMAT(5E12.6)
PUNCH40,B11,B12,B13,B14,G12
PUNCH41,A22,A23,A24,G21
41 FORMAT(12X,4E12.6)
PUNCH41,B22,B23,B24,G22
PUNCH42,A33,A34,G31
42 FORMAT(24X,3E12.6)
PUNCH42,B33,B34,G32
PUNCH43,A44,G41
43 FORMAT(36X,2E12.6)
PUNCH43,B44,G42
  B4=G42
  B3=G32-B4*B34
  B2=G22-B3*B23-B4*B24
  B1=G12-B2*B12-B3*B13-B4*B14
PUNCH4
PUNCH78,B1,B2,B3,B4
78 FORMAT(3HB1=,E12.6,3HB2=,E12.6,3HB3=,E12.6,3HB4=,E12.6)
PUNCH4
  C44=1.0/A44
  C34=-B34*C44
  C24=-B23*C34-B24*C44
  C14=-B12*C24-B13*C34-B14*C44
  C33=1.0/A33-B34*C34
  C23=-B23*C33-B24*C34
  C13=-B12*C23-B13*C33-B14*C34
  C22=1.0/A22-B23*C23-B24*C24
  C12=-B12*C22-B13*C23-B14*C24
  C11=1.0/A11-B12*C12-B13*C13-B14*C14
PUNCH45
45 FORMAT(23HDOOLITTLE BACK SOLUTION)
PUNCH46,C11,C12,C13,C14
46 FORMAT(4E12.6)
PUNCH47,C22,C23,C24
47 FORMAT(12X,3E12.6)
PUNCH48,C33,C34
48 FORMAT(24X,2E12.6)
PUNCH49,C44
49 FORMAT(36X,E12.6)
76 PUNCH77
77 FORMAT(14X,6HLENGTH,4X,3HWT1,6X,3HWT2)
  K=MSET
  DO55I=1,K,1
  DIST=I
  WT1=A*DIST**B
  WT2=1000.+B1+B2*DIST+B3*DIST*DIST+B4*DIST*DIST*DIST
PUNCH66,DIST,WT1,WT2
55 CONTINUE
66 FORMAT(10X,F10.2,F10.5,F12.5)
  4 FORMAT(4X)
  PUNCH4
  PUNCH4
  PUNCH4
  PUNCH4
  PUNCH4
  GO TO 1
  END
```


Card Format

1) A Header Card precedes the deck of data cards and sets control parameters in the program.

a. Columns 1 and 2 (Format I2) instructs the program to substitute integer values of length into fitted equations,

$$W = aL^b \tag{2}$$

$$W = b_1 + b_2L + b_3L^2 + b_4L^3 \tag{3}$$

with L = 1 to m (where m is the value typed into columns 1 and 2), and to print out these values and the values of weight for each equation.

Examples: Column 1 2

Example: 0 9 - substitutes length integers from 1 to 9 into equations

Example: 9 9 - substitutes length integers from 1 to 99 into equations

b. Columns 3, 4, and 5 are left blank.

c. Column 6 (Format I1) instructs the program to change units to inches and pounds before computation of equations (2) and (3) and their statistical parameters.

Number placed in Column	Transformation
<u>6</u>	<u>Input data-----Output</u>
1 or blank	Output in same units as input (use for data in inches-pounds)
2	millimeters-grams to inches-pounds
3	millimeters-pounds to inches-pounds
4	inches-grams to inches-pounds
5	inches-ounces to inches-pounds

- d. Column 7 is left blank
- e. Column 8 instructs the program to include or delete computation of equation (3) and print out of parameters and abbreviated Doolittle solution of equation (3).

Examples: 1. When left blank computations of equation (3) are included in output.

2. A numeral 2 in column 8 deletes above computation from output.

2) A Subheader Card instructs the program that the next card will be a header card or other than a data card. This card is used only where two or more systems of measurements are used to compute a single equation. All columns are blank on this card except a 2 must be typed in column 72.

3) Data Cards - Format (F5.2, F8.0, F10.5, F4.0)

a. Columns 1 through 5, Format (F5.2) contain values of length. The values may be in any units with or without decimal fractions. A decimal must be typed in or else it is assumed to lie between columns 3 and 4. (For inch-group where the inch-integer is the midpoint, it is suggested that the decimal be typed in column 5, which allows using the same data cards for Program II - This also applies to length measures in centimeter-integers)

Examples: Columns 1 2 3 4 5

<u>Intended Number</u>		<u>Number interpreted as:</u>
5	5	5.000000
5	5 .	5.000000
5	5	50.000000
5		Data card rejected
5	5 . 0 0 0	5.000000
0.0	0 . 0	Data card rejected

- b. Columns 6 through 13, Format (F8.0), contain the number of fish per length-group. If there is only one fish for each length-group, 1.0 must be typed in these columns. Data cards with blanks or zeros are rejected.

Example: Columns 6 7 8 9 10 11 12 13

<u>Intended number</u>								<u>Number interpreted as</u>
2						2		20.000000
2							2	2.0000000
245		2	4	5	.			245.00000
2					2	.		2.0000000
245		2	4	5				2450000.0

- c. Columns 14 through 23, Format (F10.5) contain the total weight of all fish in columns 6 through 13. Any unit of weight may be used. Unless a decimal is typed in, it is assumed to lie between columns 18 and 19. Data cards expressing trace by zero are rejected.
- d. Columns 24 through 27, Format (F4.0) contain a problem number (an identification number for the researcher) limited to 3 digits and a decimal - THIS NEED NOT BE TYPED ON DATA CARDS IF IT IS TYPED ON TRAILER CARD.
- e. The remaining columns do not enter the computation; hence any coding system or additional data may be entered in these spaces. However, it is suggested that any identification numbers be placed in columns 73 through 80.

- 4) Additional Notes: (1) This program deletes cards with zeros or blanks for values of length or weight or the number of fish from the computations; therefore, blank cards of a different color may be used to separate data cards into groups. (2) It is suggested that decimals be typed in for all values. (3) To use program III that transforms units in millimeters-grams, millimeters-pounds, inches-grams, inches-ounces, and inches-pounds

into inch-groups and pounds, columns 1, 6, 14, 28, and 31 must be blank on the data cards.

5) A trailer card instructs the computer that a problem is over and clears the memory for the next problem:

- a) Columns 1 through 5 must have 9999. typed in.
- b) Columns 6 through 23 are blank.
- c) Columns 24 through 27 must have the problem number typed in and limited to 4 digits or 3 digits and decimal.

Loading Sequence When data for one problem are all in the same units, the sequence of cards through the IBM 1620 is as follows:

- (A. Object program deck
- (B. Header card for problem 1
- 1. (C. Data Deck for problem 1
- (D. Trailer card for problem 1
- (E. Header card for problem 2
- 2. (F. Data Deck for problem 2
- (G. Trailer card for problem 2
- (H. Header card for problem 3
- 3. (I. Data Deck for problem 3
- (J. Trailer card for problem 3
- • •
- • •
- n. (ZZZC. Header card for problem n
- (ZZZD. Data Deck for problem n
- (ZZZE. Trailer card for problem n

When data for a problem are of mixed units, the sequence of cards through the the IBM 1620 is as follows:

- (A. Object program deck
- (B. Header card for problem 1 (a 2 in column 6)
- (C. Data Deck for problem 1 (Units in mm-gm)
- 1. (D. Subheader card for problem 1
- (E. Header card for problem 1 (a 3 in column 6)
- (F. Data Deck for problem 1 (units in mm-lbs)
- (G. Trailer card for problem 1

- (H. Header card for problem 2 (column 6 blank)
- (I. Data Deck for problem (units in inches-lbs)
- 2. (J. Subheader card for problem 2
- (K. Header card for problem 2 (a 5 in column 6)
- (L. Data Deck for problem 2 (units in inches-ounces)
- (M. Trailer card for problem 2

. . .
 . . .

- n. ZZZC. Header card for problem n
- ZZZD. Data Deck for problem n
- ZZZE. Trailer card for problem n

The above sequence is merely an example; any combination of different units may be used in a single problem as long as the data deck for each group of units is preceded by the correct header card as specified under card format for program I, lc. and by a subheader card.

Output

The following table is the answer sheet obtained by processing the answer cards through an IBM accounting machine. The symbol or symbols to the left of an equals sign are as the computer prints out. To the right of the equals sign, the author has chosen to place the appropriate statistical symbol or else an explanation. Numerical answers would normally appear to the right of the equals sign and would be of two forms, a number consisting of an integer and decimal fraction or else as exponential numbers of the form .324505E+04, .324505E-00, and .324505E-02, which would be interpreted as follows: 3245.05, .324505, and 0.00324505, respectively. The first eight lines of output are derived from the logarithmic forms of length and weight, whereas the next 16 lines of output are derived from the length and weight not transformed to logarithms. A capital E is used to represent the symbol sigma.

1) Printed Output

Line

- 1. SX = EX SX2 = EX² SXY = EXY
- 2. SY = EY SY2 = EY²

3. $SD = Ed^2$
4. $CORR\ COEF = r$ $AVGX = \bar{x}$ $AVGY = \bar{y}$
5. $SFNO = \text{Total No. Fish}$; $PBNO = \text{problem no.}$; $SSX2 = Ex^2$; $SSY2 = Ey^2$
6. $TOTN = \text{number of observations (n)}$; $SSXY = Exy$
7. $B = b$; $A = a$; $ALOG = \log_{10} a$
8. $STD\ ERROR = Sy.x$
9. $SUMX = EX$ $SUMX2 = EX^2$ $SUMX3 = EX^3$ $SUMX4 = EX^4$
10. $SUMX5 = EX^5$ $SUMX6 = EX^6$ $SUMY = EY$
11. $SUMXY = EXY$ $SUMX2Y = EX^2Y$ $SUMX3Y = EX^3Y$

DOOLITTLE FORWARD SOLUTION

12. A_{00} A_{01} A_{02} A_{03} A_{0y}
13. B_{00} B_{01} B_{02} B_{03} B_{0y}
14. A_{11} A_{12} A_{13} A_{1y}
15. B_{11} B_{12} B_{13} B_{1y}
16. A_{22} A_{23} A_{2y}
17. B_{22} B_{23} B_{2y}
18. A_{33} A_{3y}
19. B_{33} B_{3y}
20. $B1 = b_1$ $B2 = b_2$ $B3 = b_3$ $B4 = b_4$

DOOLITTLE BACK SOLUTION

21. C_{00} C_{01} C_{02} C_{03}
22. C_{11} C_{12} C_{13}
23. C_{22} C_{23}
24. C_{33}

LENGTH WT1 WT2 Where WT1 = weight calculated by equation (2) and
WT2 = weight calculated by equation (3)

25. 1.00 - -
26. 2.00 - -

27.	3.00	-	-
28.	4.00	-	-
29.	5.00	-	-
30.	6.00	-	-
31.	7.00	-	-
.	.	.	.
m+24	m	-	-

2) Notes on output:

a) The values of line 7 are used in the equations:

$$\log W = \log a + b \log L \quad (1)$$

$$W = aL^b \quad (2)$$

b) The values of line 20 are used in the equation:

$$W = b_1 + b_2L + b_3L^2 + b_4L^3 \quad (3)$$

c) The values in lines 1 through 6 may be used in the analysis of covariance of two length-weight relationships as follows: (this is presented merely to illustrate the exact meaning of the statistical symbols in line 1 through 8)

Source	df	Ex ²	Exy	Ey ²	b	New df	Ed ²	MS
Population 1	n ₁ -1	Ex ₁ ²	Exy ₁	Ey ₁ ²	b ₁	n ₁ -2	Ed ₁ ²	Ed ₁ ² /(n ₁ -2)
Population 2	n ₂ -1	Ex ₂ ²	Exy ₂	Ey ₂ ²	b ₂	n ₂ -2	Ed ₂ ²	Ed ₂ ² /(n ₂ -2)
Within	-	-	-	-	-	n ₁ +n ₂ -4	Ed ₁ ² +Ed ₂ ²	
Regression	-	-	-	-	-	1		
Common	n ₁ +n ₂ -2	Ex ₁ ² +Ex ₂ ²	Exy ₁ +Exy ₂	Ey ₁ ² +Ey ₂ ²		n ₁ +n ₂ -3		
Adjusted Means	-	-	-	-		1		
Total	n ₁ +n ₂ -1	Ex _T ²	Exy _T	Ey _T ²		n ₁ +n ₂ -2		

Where $Ex_T^2 = (EX_1^2 + EX_2^2) - (EX_1 + EX_2)^2 / (n_1 + n_2)$

$$Exy_T = (EXY_1 + EXY_2) - (EX_1 + EX_2)(EY_1 + EY_2) / (n_1 + n_2)$$

$$E_y \frac{2}{T} = (EY_1^2 + EY_2^2) - (EY_1 + EY_2)^2 / (n_1 + n_2)$$

Subscripts 1 and 2 refer to the population.

- d) Lines 9 through 11 form X'X matrix or A matrix and G matrix of Hader and Grandage (1958) as follows:

Hader and Grandage Symbols

$$X'X = A \qquad G$$

a ₀₀	a ₀₁	a ₀₂	a ₀₃	g ₀
	a ₁₁	a ₁₂	a ₁₃	g ₁
		a ₂₂	a ₂₃	g ₂
			a ₃₃	g ₃

or in our symbols

$$X'X = A \qquad G$$

n	EX	EX ²	EX ³	EY
	EX ²	EX ³	EX ⁴	EXY
		EX ⁴	EX ⁵	EX ² Y
			EX ⁶	EX ³ Y

- e) Lines 12 through 19 are Hader and Grandage (1958) symbols for the Forward Abbreviated Doolittle solution of the X'X matrix or a matrix.

- f) Lines 21 through 24 are Hader and Grandage's symbols for the inverse X'X or C matrix.

Note: A test of the accuracy of the computations would be to calculate the identity matrix, I, by AC = I which should be:

$$\begin{matrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{matrix}$$

The degree with which the diagonal values approach unity would give a check on the accuracy.

g) Lines 25 through $m + 24$ are in tabular form and are the values of weight determined from equations (2) and (3) at integer intervals of length 1 to m . Weights under the heading WT1 are derived from equation (2). Weights under the heading WT2 are derived from equation (3) and have a factor of 1000.0 added to each weight. If the option is taken to delete computation of equation (3), then all values under WT2 will be numerically equal to 1000.0 and lines 9 through 24 will be deleted from the output. When observational values are missing from either end or the middle of the range over which the output is computed, then equation (3) often gives negative values for these missing values. These must be disregarded and only the section of the calculated curve which fits the empirical data should be used.

h) Line $m + 24$, the value of m is the value typed into columns 1 and 2 of the header card.

i) Occasionally the answers are too large to be printed in the allotted space. In this case the space where the answer should be is blank and the next line has the caption: ERROR F8, which is followed by the answer in exponential form.

j) The author found that for some species of fish a single length-weight equation failed to adequately describe the length-weight relationship over the entire range in length; therefore, it was necessary to compute two or more equations for some species. The point (or points) chosen for separation of data were those where the increment of increase or decrease in the condition index changed in magnitude.

Program II

Title: Length-Weight Relationships II, IBM 1620, Fortran/Format

This program is useful where fish are grouped in class intervals according to length. The program computes average empirical weight, maximum and minimum weights, condition index, and number of fish for each interval. The results of these com-

putations are printed in tabular form. The condition index is computed from the average empirical weight for each interval by the formula,

$$C = \frac{WX10^5}{L^3} \quad (4)$$

Data cards are processed at the rate of 120 per minute, with punch out requiring an additional minute. Accuracy is limited only by the accuracy of the of the data; 5470 storage positions are utilized.

Restrictions: Number of size intervals must be equal to or less than 99. The class mark of each interval must be an integer. Input and output are on cards with one data card required for each observation.

Card Format

- 1) Header Card
 - a. Same as for (a) under Program I.
 - b. All other spaces are blank or same header card as used for Program I may be used.
- 2) Data Cards
 - a. Decimal point must be in column 5. A two-digit length would be typed in columns 3 and 4 with the decimal in column 5. A one digit length would be typed in column 4 with the decimal in column 5. The maximum length that may be entered is the two-digit number 99.
 - b. Same as for (b) under Program I.
 - c. " " " (c) " " "
 - d. " " " (d) " " "
 - e. " " " (e) " " "
- 3) Trailer Card - same as for Program I
- 4) Additional notes

For data in units other than inch-groups and pounds, data cards should be typed in same units of measure as field data and used with Program I first. Then Program III may be used to transform these data cards to data cards in inch-groups and pounds which may be used with Program II.

Source Program

```

DIMENSIONW(99),F(99),A(99),C(99),SI(99),BI(99),SM(99),AV(99),B(99),S(99)
16 READ 13,LEVJ
13 FORMAT(I2)
DO 15 J=1,99
W(J)=0
F(J)=0
A(J)=0
SI(J)=0
AV(J)=0
BI(J)=0
SM(J)=100000.
C(J)=0
B(J)=0
S(J)=0
15 CONTINUE
2 READ 99,J,FNO,TOTWT,PBND
99 FORMAT(I4,1X,F8.0,F10.5,F4.0)
IF(J-8888)20,14,14
20 IF(J)21,2,21
21 IF(FNO)22,2,22
22 IF(TOTWT)23,2,23
23 W(J)=W(J)+TOTWT
F(J)=F(J)+FNO
AV(J)=TOTWT/FNO
IF(AV(J)-SM(J))80,80,81
80 SM(J)=0
SM(J)=AV(J)
GO TO 83
81 IF(AV(J)-BI(J))83,82,82
82 BI(J)=0
BI(J)=AV(J)
GO TO 83
83 AV(J)=0
GO TO 2
14 PUNCH 6,PBND
6 FORMAT(5HPBND=,F8.0)
PUNCH 9
9FORMAT(4HLENG,3X,7HND FISH,4X,6HAVG WT,5X,3HC F,6X,5HMAXWT,7X,5HMINWT)
DO 7 J=1,LEVJ
IF(W(J))11,8,11
11 A(J)=W(J)/F(J)
SI(J)=J
C(J)=(A(J)*10.**5)/SI(J)**3
B(J)=BI(J)
S(J)=SM(J)
PUNCH 10,SI(J),F(J),A(J),C(J),B(J),S(J)
10 FORMAT(F4.0,2X,F8.0,2X,F12.8,2X,F8.2,2X,F12.8,2X,F12.8)
8 CONTINUE
7 CONTINUE
PUNCH4
PUNCH4
PUNCH4
PUNCH4
PUNCH4
4 FORMAT(4X)
GO TO 16
END

```

Sequence of cards through the IBM 1620 is the same as for Program I when all data are in the same units. If the cards for an inch-group are arranged so that average weights for fish of that inch-group are in descending order for maximum to minimum, then no maximum weight is computed. Where a great deal of data for each inch-group is processed, it is highly unlikely that the cards would be arranged in such an order. However, to avoid this computational error, it is necessary only to place one data card for each inch-group ahead of one that has a weight of greater magnitude.

Output

The output is in tabular form. The first column of the table is the length integers from 1 to m (where m is same as in Program I). The second column is the total number of fish for each length-group (length integer). The third column is the average weight for each length-group. Column four is a condition index derived from the following equation:

$$C = \frac{WX10^5}{L^3} \quad (4)$$

where W is equal to the average weight and L is the length-integer for the same length-group. Column five is the maximum weight for each length-group, and column six is the minimum weight.

PROGRAM III

Title: Program to Transform Data to Inch-Groups and Pounds.

This program is used to make new data cards from data cards in units other than inch-groups and pounds. The program transforms millimeters-grams, inches-grams, millimeters-pounds, and inches-ounces to inch-groups and pounds. Then these new data cards may be used with Program II.

This program utilizes the same header card, subheader card, and data cards as Program I. No trailer card is used. However, the data cards must have columns 1, 6, 14, 28, and 31 blank.

Source Program

```

1 READ 2,LSET
2 FORMAT(4X,I2)
4 READ 99,SIZE,FNO,TOTWT,K,I,KI
99 FORMAT(1X,F4.0,1X,F7.0,1X,F9.5,5X,I1,2X,I2,38X,I1)
   IF(KI-1)7,7,1
7 IF(LSET-1)105,105,100
100 IF(LSET-3)106,107,101
101 IF(LSET-5)108,109,105
105 WY=TOTWT
   WX=SIZE
   GO TO 6
106 WX=SIZE/25.4
   WY=TOTWT/453.59
   GO TO 6
107 WX=SIZE/25.4
   WY=TOTWT
   GO TO 6
108 WY=TOTWT/453.59
   WX=SIZE
   GO TO 6
109 WY=TOTWT/16.0
   WX=SIZE
6 WXL=1.5
   DO 70 J =1,99
   IF(WX-WXL)65,66,66
65 X=WXL-.5
   GO TO 88
66 WXL=WXL+1.
70 CONTINUE
88 PUNCH 3,X,FNO,WY,K,I
3 FORMAT(F5.0,F8.0,F10.5,4X,I2,1X,I3)
   X=0
   WX=0
   WY=0
   FNO=0
   K=0
   I=0
   KI=0
   GO TO 4
END

```

Data cards of different units can be placed in sequence through the IBM 1620, as shown in the table for the sequence of data of mixed units for Program I. The program will not process data cards for fish longer than 99 inches.

LITERATURE CITED

Hader, R. J. and A. H. E. Grandage. "Simple and Multiple Regression Analysis", Experimental Designs in Industry, John Wiley Company, 1958.

